********Table Q.1 - Indonesia********

eststo clear


*Load data and create differences
cd "$reploc/data/created_data"
use indo_full, clear

sort pidlink ifls
drop if ifls==.
keep if panel==1

keep rbuc*  gambleaverse ifls pidlink inflation imp_age panel biryear_impl grbrprovAvg_lam0 grbrprovSD_lam0 grbrprovAvg_lam0co? grbrprovSD_lam0co? birprov_impl


reshape wide rbuc*  gambleaverse inflation imp_age panel grbrprovAvg_lam0 grbrprovSD_lam0 grbrprovAvg_lam0co? grbrprovSD_lam0co?, i(pidlink) j(ifls)


local varlist1 rbuc rbucB rbuc_nga rbuc_garefined rbucgaref1a rbucgaref2a rbucgaref3a rbucgaref4a gambleaverse inflation grbrprovAvg_lam0 grbrprovSD_lam0 grbrprovAvg_lam0co2 grbrprovSD_lam0co2 grbrprovAvg_lam0co3 grbrprovSD_lam0co3 


foreach var in `varlist1'  {
gen `var'dif=.
replace `var'dif=`var'5-`var'4
}

*Generate alternate clusters
*5-year by state of birth
egen fivyearclust=cut(biryear_impl), at(1915(5)2020) icodes
egen fivcoh_prov_clust=group(fivyearclust birprov_impl)

*10-year by state of birth
egen tenyearclust=cut(biryear_impl), at(1915(10)2020) icodes
egen tencoh_prov_clust=group(tenyearclust birprov_impl)

*15-year by state of birth
egen fifteenyearclust=cut(biryear_impl), at(1915(15)2020) icodes
egen fifteencoh_prov_clust=group(fifteenyearclust birprov_impl)

*Regressions
local x rbuc_garefined
local rhsvari oldindoonly
local rhsvaruse _lam0co3dif
local usei grbr

drop if `x'dif==.

*Column 1
eststo:  reg `x'dif  `usei'provAvg`rhsvaruse' `usei'provSD`rhsvaruse'  inflationdif, vce(cluster fivcoh_prov_clust)

*Column 2
eststo:  reg `x'dif  `usei'provAvg`rhsvaruse' `usei'provSD`rhsvaruse'  inflationdif, vce(cluster tencoh_prov_clust)

*Column 3
eststo:  reg `x'dif  `usei'provAvg`rhsvaruse' `usei'provSD`rhsvaruse'  inflationdif, vce(cluster fifteencoh_prov_clust)

cd "$reploc/results"	
esttab using TableQ1_Indonesia.tex, star(* 0.1 ** 0.05 *** 0.01) b(2) se(2) replace
